

Programando con VBA

Abra el libro de Excel llamado "Práctica con VBA". Ubíquese en la hoja de "Asientos".



Ya hemos programado el ícono de **buscar cuenta**, que nos permite encontrar el código de una cuenta en la introducción de asientos. Ahora programaremos el ícono **Guardar Asiento**. Lo que se quiere que este ícono haga cuando demos clic sobre él es que el asiento que realizamos lo copie en la hoja que se llama "Registros". Tenga el cuidado de no cambiar los nombres de las hojas del libro. El código lo escribiremos suponiendo que los nombres de las hojas del libro se mantienen igual.

Primero introduciremos el siguiente asiento. Para encontrar los códigos de las cuentas, utilice el ícono de búsqueda. Tenga en cuenta que para utilizar este ícono es necesario que usted primero coloque el cursor en la celda donde desee que se pegue el código.

En el campo donde dice **N° de Asiento**, introduzca el número 1. Este campo se actualizará automáticamente en los asientos siguientes.

Datos del Asiento:

FECHA	CÓDIGO DE CUENTA	NOMBRE DE LA CUENTA	DESCRIPCIÓN	DÉBITOS	CRÉDITOS
01-dic-12		CAJA	Apertura de la cuenta	1,500.00	
01-dic-12	Usted	FONDO FIJO DE CAJA CHICA	Apertura de la cuenta	5,000.00	
01-dic-12	deberá	BANCOS	Apertura de la cuenta	20,000.00	
01-dic-12	buscar	INVENTARIOS	Apertura de la cuenta	100,000.00	
01-dic-12	los	TERRENOS	Apertura de la cuenta	150,000	
01-dic-12	códigos	EDIFICIOS	Apertura de la cuenta	250,000	
01-dic-12	de las	PROVEEDORES	Apertura de la cuenta		75,000.00
01-dic-12	cuentas.	ACREEDORES BANCARIOS	Apertura de la cuenta		200,000
01-dic-12		CAPITAL INICIAL	Apertura de la cuenta		251,500

Verifique en la parte superior derecha que el asiento esté cuadrado, es decir que el **Total Débitos** y **Total Créditos** sean iguales. La **diferencia** debe estar en cero.

El asiento está listo para ser trasladado a la hoja que se llama "Registros". Para lograr esto, comenzaremos a programar el ícono "Guardar Asiento".

Hagamos primero un recuento:

Vaya a la ficha **Programador** y dé clic en el ícono de **Visual Basic**. En la clase pasada, habíamos creado en el editor de Visual Basic los módulos llamados: **menu_principal** y **menu_asientos**. En el primero, escribimos el

código que nos permitió agregar acción a los íconos superiores que utilizamos para navegar por las hojas del libro. En **menu_asientos**, estamos escribiendo el código de todos los íconos que están en la hoja **Asientos**.

En este módulo (menu_asientos) ya escribimos el código del ícono Buscar Cuenta, ahora, debajo de este código escribiremos el código para trasladar el asiento a la hoja **Registros**.

Iniciamos:

Asegúrese de tener abierto la hoja del módulo menu_asientos. Si aún no lo ha hecho, busque ese módulo en la parte izquierda de la ventana de Visual Basic, y dé doble clic sobre él. Ubíquese debajo del código que ya escribimos en este módulo y a continuación siga estas instrucciones:

Indique el nombre de la nueva macro. La llamaremos **guardar_asiento**.

`Sub guardar_asiento()`

Ahora procederemos a agregar algunas acciones que nos garanticen que el asiento que vamos a registrar no vaya con errores. Trataremos de incluir el manejo de los errores más básicos para no hacer complejo el código. La práctica lo capacitará a usted para escribir posteriormente códigos más complejos.

Una de las primeras cosas es que no debemos guardar un asiento que no esté debidamente cuadrado. Cuando un asiento no está cuadrado, aparece el valor de la diferencia en la celda llamada “**diferencia**”, es decir, el asiento está descuadrado cuando el **valor de diferencia es distinto de cero**. Por tanto, incluyamos un código para tratar este error. Lo haremos de la siguiente manera.

Escribamos a continuación lo siguiente en el editor del Visual Basic, debajo de la línea `Sub guardar_asiento()`.

```
' -----  
' Verificando si el asiento está cuadrado  
' -----
```



Lo que está en verde es un comentario, que permite las futuras revisiones del código. El comentario es una breve descripción acerca de lo que hace el código que se encuentra debajo de él. Cuando escriba un comentario, inicie la línea con una comilla simple ('). No incluya una comilla doble (") porque le dará un mensaje de error. En su teclado deberá localizar la comilla simple para poder incluir comentarios descriptivos en su código.

Debajo del comentario le diremos a Excel que **si el asiento está descuadrado, que envíe un mensaje de alerta que diga: “El asiento no puede registrarse porque está descuadrado”** y adicionalmente le indicaremos a Excel que si el asiento no está cuadrado, que no realice el registro. Para lograr esto escribiremos lo siguiente:

1. `If Range("diferencia") <> 0 Then`
2. `MsgBox "El asiento no puede registrarse porque está descuadrado", vbInformation, "Asiento descuadrado"`
3. `End`
4. `End If`

Interprete en la siguiente hoja el significado de estas líneas de código.

Significado de las líneas anteriores:




3. **End** (finalizar). Esta instrucción le indica a Excel que detenga todas las macros y que no realice ninguna acción. En este caso, el asiento no se registra.
4. **End If** (Final del bloque Si). Esta instrucción sirve para indicarle a Excel que hasta ahí llegan las instrucciones del bloque condicional If.

El bloque condicional If funciona igual que la función = Si (prueba_logica, parte_verdadera, parte_falsa) de Excel. En este código, la función If, lo que hace es comprobar si el asiento está descuadrado, si esta condición resulta verdadera, entonces lanzará un cuadro de mensaje indicando la razón por la cual el asiento no se guardará.

Probemos nuestro código:

Intente guardar un asiento descuadrado. Para eso escriba en la cuenta de capital una cantidad distinta, por ejemplo 200,000 con esto el asiento queda descuadrado. Haga clic derecho sobre el icono **Guardar Asiento** y asígnele la macro **guardar_asiento**. Luego, haga clic sobre el mismo ícono y deberá aparecer lo siguiente:



Si no aparece el cuadro de mensaje, entonces revise el código porque posiblemente escribió algún error.

Ahora bien, ocurre lo siguiente. Si el primer renglón de los asientos lo dejamos en blanco, se producirían errores al momento del guardado, así que procederemos a incluir un tratamiento para todos los posibles eventos que no deseamos que ocurran en el registro de la información contable.

Debajo de **End If** (pero no debajo de End Sub) escribiremos el siguiente comentario:

```
'-----  
' Comprobación de campos vacíos y errores en el registro  
'-----
```

1. `If Range("A15") = "" Or Range("B15") = "" Or Range("C15") = "" _
Or Range("D15") = "" Or Range("E15") = "" Then`
2. `MsgBox "No se puede realizar un registro cuando el primer renglón" _
& Chr(10) & "del mismo se encuentra vacío. Corrija el error e intente nuevamente" _` ←
3. `End`
4. `End If`


En la línea N° 1, `If Range ("A15") = ""` lo que se le está pidiendo a Excel es que compruebe si la **Celda A15** está vacía o está en blanco. Asimismo en la misma línea se le pide al programa que verifique si las demás celdas (B, C, D y E) del mismo renglón 15 están vacías. Aquí se utiliza el conector **Or** que significa **O**. Si Excel encuentra alguna de esas celdas con un valor nulo (en blanco), entonces Excel desplegará un mensaje de error y no permitirá realizar el registro hasta que se rellenen todos los campos del renglón 15.

En la línea del código donde aparece la instrucción **MsgBox**, observe que cuando la línea de código es un poco larga, conviene seguir escribiendo en la línea siguiente, pero para que no se rompa el vínculo con la instrucción, antes de presionar **Enter** para continuar en la siguiente línea, es necesario dar un espacio seguido de un guion bajo y luego continuar en la siguiente línea.

Adicionalmente, si en los cuadros de mensajes de alerta se desea incluir un texto bastante extenso y conviene que el texto se muestre en más de un renglón, entonces entre las líneas de texto de cada de cada renglón debe incluirse la instrucción **& Chr(10) &**. Los textos del mensaje de alerta deben incluirse entre comillas.

Por ejemplo: El texto "Hoy es lunes y mañana es martes". En un cuadro de mensaje se presentará en una sola línea. Pero "Hoy es lunes" & Chr(10) & " y mañana es martes". En un cuadro de mensajes se presentará en dos líneas.

Después de escribir el código, pruebe a dejar en blanco cualquiera de las celdas del primer renglón, excepto las celdas de la columna donde va el nombre de la cuenta, ya que estas celdas están con fórmulas. Después haga clic sobre el ícono **Guardar Asiento**. Deberá aparecer lo siguiente:



Deberíamos incluir un código que permitiera impedir que se registren asientos cuando haga falta alguno de los datos como la fecha, el código de la cuenta, la descripción o cualquiera de los otros datos. Sin embargo, esta tarea se volvería bastante compleja. Así que bastará con tener cuidado de revisar si toda la información necesaria está completa antes de guardar el asiento.

Procederemos ahora a escribir el código que permitirá realizar el registro del asiento.

Debajo de la última instrucción que escribimos (End If) escribiremos lo siguiente:

```
'-----  
'Se realiza el registro del asiento  
'-----
```

Primero debemos determinar el renglón en el cual se pegará el asiento que vamos a registrar. Esto es debido a que en cada asiento, es posible que hayan otros asientos registrados y debemos asegurarnos que el nuevo asiento se pegará debajo de los que ya están. Para esto, debemos contar el número de líneas que están ocupadas en la hoja **Registros**. Después de esto, podemos indicarle a Excel que escriba el nuevo contenido en la siguiente línea que esté disponible.

En este punto será de mucha ayuda la utilización de variables dinámicas. Una variable dinámica es un elemento que puede adquirir un valor dinámico y ese valor podemos utilizarlo en alguna operación. Para declarar una variable, se introduce primero el tipo de variable, seguido del nombre de la variable y el tipo de datos que almacenará esa variable. El tipo de variables que utilizaremos es el Dinámico (**Dim**) y los tipos de datos que almacenan las variables son: **String** (texto), **Integer** (entero), **Long** (largo), **Double**(doble). Hay otros tipos, pero estos son los más comunes que emplearemos.

Declararemos la variable **A**, la cual utilizaremos para llevar el control de las líneas ocupadas en la hoja **Registros**. Como puede darse cuenta, la cantidad de registros que puede almacenarse en una hoja de trabajo puede llegar a ser un poco grande, por tanto, a esta variable le asignaremos el tipo de datos **Long**.

Declaremos la variable:

```
Dim A As Long 'Contador de la cantidad de registros que se han agregado en la hoja Registros  
Dim B As Integer 'Contador de la cantidad de líneas del nuevo asiento que se va a agregar a Registros.
```

La variable **B** la utilizaremos para llevar un control de la cantidad de renglones que contiene el asiento nuevo que se agregará a la hoja de registros. Como suponemos que un asiento no tiene una cantidad tan grande de renglones, entonces a esta variable **B** se le ha asignado el tipo de datos **Integer**.

Ahora, calcularemos los valores de ambas variables de la siguiente manera:

```
A = WorksheetFunction.CountIf (Worksheets ("Registros").Range ("A : A"), "<>" & "")  
B = WorksheetFunction.CountIf (Worksheets ("Asientos").Range ("A15", "A" & Rows.Count), "<>" & "")
```

Explicación de las dos líneas anteriores:

A = WorksheetFunction.CountIf (Worksheets ("Registros").Range ("A : A"), "<>" & "")

El valor de la variable será igual a

Indica que se hará uso de una función de Excel: COUNTIF. Equivalente a =CONTAR.SI ()

Rango donde se encuentran los datos que se desean contar (en este caso toda la columna A de la hoja Registros.

Que se cuenten las celdas que no están vacías.

La **función Countif** del VBA es la equivalente a la función **=CONTAR.SI ()** de Microsoft Excel. Con esta función se cuenta solo algunos datos que cumplen con un criterio. Para nuestro ejemplo, únicamente estamos contando aquellas Celdas de la **Columna A** de la hoja **Registros** que no están en blanco.

B = WorksheetFunction.CountIf (Worksheets ("Asientos").Range ("A15", "A" & Rows.Count), "<>" & "")

Para el segundo caso, el significado es similar, pero hay una variante en el rango en el que se encuentran los datos que se contarán:

En el conteo no se incluirán los datos de toda la columna A (A:A), sino que el conteo iniciará a partir de la Celda "A15" en adelante. La Expresión en adelante, indica hasta el último renglón de la columna A. Para esto se emplea la expresión Rows.Count que significa "Cuenta de filas", si la cuenta de filas es por ejemplo 100, la expresión "A" & Rows.Count significaría "A100".

Ahora que ya tenemos definida las variables que necesitaremos, podemos comenzar a registrar algunos datos. En el área donde se pegará el asiento, al inicio de cada registro colocaremos un pequeño título que diga: Asiento #: (y el número del asiento que se está registrando). Para eso escribimos el siguiente código:

Worksheets ("Registros").Range ("A" & A + 1) = "Asiento #: " & Worksheets ("Asientos").Range ("no.asiento")

En el código anterior, en la referencia al **Range()**, la **A** que está entre comillas se refiere a **la columna** de la hoja Registros y la **A** que está sin comillas se refiere a al **valor de la variable** que definimos anteriormente. En este caso, la **Variable A** será igual al número de filas que ya están ocupadas en la hoja Registros. Por tanto, **A + 1** se refiere a la fila que está después de la última fila ocupada en la hoja registros.

Worksheets("Registros").Range("A" & A + 1).Font.Bold = True

La primera parte de la línea anterior hace referencia a la celda donde se escribirá el título del asiento y la segunda parte (.Font.Bold = True), le dice a Excel que le aplique formato de Negrita al título del asiento. El título quedaría parecido a lo siguiente: **Asiento #: XX**.

Después, seleccionamos los datos del asiento que se va a copiar. El asiento inicia en la **Celda "A15"** de la hoja **Asientos** y llega hasta la columna "F", pero en este punto necesitamos indicar el índice de la columna "F" para delimitar la celda. ¿Cuál será? ¿La "F15"? ¿La "F16"?... ¿Cuál? Para esto utilizaremos la Variable **B** que ya definimos con anterioridad y que se encarga de contar las filas del asiento que se va a registrar. El código quedaría de la siguiente manera:

Worksheets ("Asientos").Range ("A15", "F" & B + 14).Copy _
Destination:=Worksheets ("Registros").Range("B" & A + 2)

En la primera línea, la primera parte: `Worksheets ("Asientos")` hace referencia a la hoja **Asientos**. La segunda parte de la misma línea: `.Range ("A15", "F" & B + 14)` le dice a Excel que la sección donde se encuentra el asiento a copiar es la que está desde la **Celda A15** hasta la **F** y el **valor de B más las 14** líneas situadas en la parte superior de asiento que no se están utilizando. Para nuestro ejemplo, el asiento que queremos guardar tiene 9 líneas. Por tanto, el valor de **B = 9**. Recuerde que **B** lo que hace es contar cuántas líneas tiene el asiento que estamos registrando. Entonces si a **B** le sumamos las **14 líneas** que están antes de la línea 15 que es donde comenzamos a escribir el asiento (recuerde que el asiento lo iniciamos en la "A15") tendremos como resultado 23 ($9 + 14 = 23$). Por tanto la expresión `"F" & B + 14 = F23`. En otras palabras, Excel copiará solamente el área desde la **Celda A15 hasta la F23**. Seguidamente el comando `.Copy` le da la orden a Excel de que el rango al que se hace referencia lo copie para luego realizar alguna operación con el mismo.

FECHA	CÓDIGO DE CUENTA	NOMBRE DE LA CUENTA	DESCRIPCIÓN	DÉBITOS	CRÉDITOS
01-dic-12	1101	CAJA	Apertura de la cuenta	1,500.00	
01-dic-12	1102	FONDO FIJO DE CAJA CHICA	Apertura de la cuenta	5,000.00	
01-dic-12	1103	BANCOS	Apertura de la cuenta	20,000.00	
01-dic-12	1110	INVENTARIOS	Apertura de la cuenta	100,000.00	
01-dic-12	1201	TERRENOS	Apertura de la cuenta	150,000.00	
01-dic-12	1202	EDIFICIOS	Apertura de la cuenta	250,000.00	
01-dic-12	2101	PROVEEDORES	Apertura de la cuenta		75,000.00
01-dic-12	2104	ACREEDORES BANCARIOS	Apertura de la cuenta		200,000.00
01-dic-12	3101	CAPITAL INICIAL	Apertura de la cuenta		251,500.00

Celda "A15" hasta la Celda "F23"

En la segunda línea: `Destination:=Worksheets ("Registros")` se le indica a Excel que los datos se copiarán en la hoja **"Registros"** se le antepone la expresión `Destination:=` esta expresión va inmediatamente después de la orden `.Copy`, si por algún motivo no se puede escribir en la misma línea, lo recomendable es que después del comando `.Copy` se deje un espacio seguido de un guion bajo y en la siguiente línea se escribe el `Destination`. Posteriormente en la segunda parte de la misma línea, la expresión `.Range ("B" & A + 2)` indica que los datos se pegarán desde la celda **B** + el **Valor de la Variable A + 2**. Anteriormente utilizamos la variable **A** para indicar el renglón donde pegaríamos el título **Asiento #: XX**. En esa ocasión utilizamos la expresión `A + 1`. Pero si seguimos empleando `A + 1`, el asiento se pegará encima el título **Asiento #: XX**, por tanto emplearemos la expresión `A + 2`, es decir, la línea después de `A + 1` que es donde se encuentra el título del asiento que registraremos.

Con esto hemos logrado que Excel copie y pegue el asiento de la hoja **Asientos** a la hoja **Registros**. Si probamos a ejecutar el código en estos momentos obtendremos el siguiente resultado en la hoja **Registros**:

	A	B	C	D	E	F	G
1	N° Asiento	FECHA	CÓDIGO DE CUENTA	NOMBRE DE LA CUENTA	DESCRIPCIÓN	DÉBITOS	CRÉDITOS
2	Asiento #: 1						
3		01-dic-12	1101	CAJA	Apertura de la cuenta	1,500.00	
4		01-dic-12	1102	FONDO FIJO DE CAJA CHICA	Apertura de la cuenta	5,000.00	
5		01-dic-12	1103	BANCOS	Apertura de la cuenta	20,000.00	
6		01-dic-12	1110	INVENTARIOS	Apertura de la cuenta	100,000.00	
7		01-dic-12	1201	TERRENOS	Apertura de la cuenta	150,000.00	
8		01-dic-12	1202	EDIFICIOS	Apertura de la cuenta	250,000.00	
9		01-dic-12	2101	PROVEEDORES	Apertura de la cuenta		75,000.00
10		01-dic-12	2104	ACREEDORES BANCARIOS	Apertura de la cuenta		200,000.00
11		01-dic-12	3101	CAPITAL INICIAL	Apertura de la cuenta		251,500.00
12							

Obsérvese que la columna en la que se encuentra el N° de Asiento, aparece en blanco. Esto es porque el número de asiento no lo incluimos directamente en el asiento como tal, sino que aparece como un campo en la parte superior de la fecha. Así que tendremos que emplear un método para agregar a cada registro el número del asiento.

Aprenderemos a emplear en esta sección el método **For...To...Next** este método permite ejecutar un código con un cierto número de repeticiones. Para nuestro caso, como en cada línea del asiento se requiere registrar el número del asiento, esto es algo repetitivo para el número de filas del asiento.

Para emplear el método **For...To...Next** se requiere definir dos variables. Una de inicio del procedimiento y la otra que marca el fin del mismo.

La variable de inicio la llamaremos *i* y la de fin la llamaremos *j*. Así que declaremos las variables

Dim i As Integer

Dim j As Integer

Ahora procederemos a calcular el valor para las dos variables. Para el caso de *i* tomaremos como base el valor de la **variable A**. Esto es así por la siguiente razón. Recordemos un poco. La variable **A** lleva el control de los registros que se han agregado en la hoja **Registros**. Esto lo hacemos mediante el conteo de las celdas en la **columna A** de la hoja **Registros** que están ocupadas (que no están en blanco). Si le echamos un vistazo a la imagen anterior comprenderemos mejor esto.

	A	B	C	D
1	N° Asiento	FECHA	CÓDIGO DE CUENTA	NO
2	Asiento #: 1			
3		01-dic-12	1101	CAJA
4		01-dic-12	1102	FONDO FIJO
5		01-dic-12	1103	BANCOS
6		01-dic-12	1110	INVENTARIO
7		01-dic-12	1201	TERRENOS
8		01-dic-12	1202	EDIFICIOS
9		01-dic-12	2101	PROVEEDOR
10		01-dic-12	2104	ACREEDORES
11		01-dic-12	3101	CAPITAL INIC
12				

No necesitamos ver toda la imagen. Después de registrar el asiento, los datos quedarían así como muestra la imagen. Recordemos que el valor de la **variable A** se ejecutó antes de realizar el registro del asiento y antes de escribir el título

Asiento#: XX. Si nos fijamos bien, antes de que se escribiera el título **Asiento #: XX** la única celda de la **columna A** que estaba ocupada era la **A1**, por tanto, al ejecutar el conteo de las celdas ocupadas el valor de **A** sería igual a 1. Para escribir el título del asiento (**Asiento #: XX**) utilizamos la expresión **A + 1** (**ver**) Si sustituimos la **variable A** por su valor (1), la expresión **A+1** es igual a **1 + 1 = 2**. Es por eso que el título del Asiento se

escribe en la **Celda A2**. Posteriormente cuando mandamos a copiar el asiento, le dimos la instrucción a Excel que el Destination quedara así: **Destination:=Worksheets ("Registros").Range ("B" & A + 2)**. En esta línea, la expresión del **Range ("B" & A + 2)** se interpreta de la siguiente manera: "**B**" es la referencia de la columna donde se pegará el asiento, pero esta "**B**" va unida mediante el nexo **&** a la expresión **A + 2**, la cual si sustituimos la **variable A** por su valor 1 como hicimos anteriormente tendríamos la siguiente referencia:

B & 1 + 2, es decir, **B3**. Si observamos el área donde se comenzó a pegar el asiento es exactamente esa misma celda la **B3**. Con esto queda entendido que el valor de **A** hasta ahorita es igual a 1. El Número del asiento

debemos iniciar a escribirlo a la izquierda de la celda B3, donde está la primera línea del asiento, es decir, debemos escribirlo en la celda A3. Pero la Celda A3 no será siempre esa misma, sino que variará de acuerdo a los asientos que vayamos introduciendo, por tanto, la mejor forma de encontrar esa celda en la que debemos iniciar a escribir los números de asientos sería mediante cálculos. En este punto es donde utilizaremos la *i*, debería iniciar por el número 3 si combinamos “Ai” estaríamos diciendo A3. Para conseguir esto podríamos establecer el valor de *i* utilizando la siguiente ecuación: $i = A + 2$. Recordemos que A vale 1 en este asiento. Así que $i = 1 + 2$, por tanto $i = 3$. Cuando combinemos “A” & *i* obtendríamos como resultado la referencia de A3.

Dejemos este punto en memoria y ahora analicemos un poco sobre el valor de la segunda variable que sería *j*. Para determinar el valor de esta variable tomaremos como base el valor de la variable B que ya definimos anteriormente. Recordemos que B lleva el control de la cantidad de líneas que contiene el asiento que estamos registrando. Para este asiento, la cantidad de líneas es de 9. Pero este 9 no es permanente, habrá asientos que tendrán más o menos de 9 renglones. Por tanto, lo más recomendable es utilizar B en vez de 9, ya que el valor de B cambiará conforme el asiento que se introduzca. Por tanto podríamos establecer el valor de *j* con la siguiente ecuación: $j = A + 1 + B$. Esto daría como resultado 11. Es decir, después de la celda A2 (A+1), registrar 9 renglones (los del asiento que estamos registrando).

Si no ha comprendido esta explicación, se sugiere leerla dos o tres veces más para continuar con la siguiente parte. Después de declarar las variables *j* e *i* y con las explicaciones realizadas, podemos escribir lo siguiente:

$$j = A + 1 + B$$

Ahora escribimos el inicio del procedimiento de registro del número de asientos mediante la siguiente expresión:

$$\text{For } i = A + 2 \text{ To } j$$

Esta expresión significa: Para *i* utilizar los valores desde A + 2 (es decir desde 3 para este caso) hasta (es el significado de To) *j* (que vale 11 para este caso). Luego dentro del bloque For escribiremos lo siguiente:

Worksheets (“Registros”).Range (“A” & i) = Worksheets (“Asientos”).Range (“no.asiento”)

Next i

Next i indica que el procedimiento se repita hasta que *i* alcance el valor de *j*, cuando esto ocurra, el procedimiento se detendrá. Como vimos *j* sería igual a 11 para este asiento, por tanto, el último número de asiento se escribiría en la celda “A” & *i*, como en este punto el valor de *i* será igual a *j*, entonces la celda final sería la “A11”. Si se fijan en la imagen, el asiento finaliza en el renglón 11.

Si probamos el código hasta aquí obtendríamos lo siguiente:

	A	B	C	D	E	F	G
1	N° Asiento	FECHA	CÓDIGO DE CUENTA	NOMBRE DE LA CUENTA	DESCRIPCIÓN	DÉBITOS	CRÉDITOS
2	Asiento #: 1						
3	1	01-dic-12	1101	CAJA	Apertura de la cuenta	1,500.00	
4	1	01-dic-12	1102	FONDO FIJO DE CAJA CHICA	Apertura de la cuenta	5,000.00	
5	1	01-dic-12	1103	BANCOS	Apertura de la cuenta	20,000.00	
6	1	01-dic-12	1110	INVENTARIOS	Apertura de la cuenta	100,000.00	
7	1	01-dic-12	1201	TERRENOS	Apertura de la cuenta	150,000.00	
8	1	01-dic-12	1202	EDIFICIOS	Apertura de la cuenta	250,000.00	
9	1	01-dic-12	2101	PROVEEDORES	Apertura de la cuenta		75,000.00
10	1	01-dic-12	2104	ACREEDORES BANCARIOS	Apertura de la cuenta		200,000.00
11	1	01-dic-12	3101	CAPITAL INICIAL	Apertura de la cuenta		251,500.00
12							

Obsérvese en la imagen, el número de asiento agregado. Con esto hemos conseguido que la información del asiento se registre correctamente. Si usted ha ido ejecutando el código o probándolo, se recomienda que en la

hoja de registro vaya borrando la información que se va agregando porque el código aún no está completo y la información se agregaría de forma desordenada.

Después de registrar el asiento, debemos limpiar el área de trabajo, nuestra área de trabajo es la hoja de asientos. Esta debe quedar limpia para seguir creando nuevos asientos, por tanto procederemos a limpiar esa área de la siguiente manera:

```
Worksheets ("Asientos").Range ("A15", "B" & B + 14).ClearContents
```

```
Worksheets ("Asientos").Range ("D15", "F" & B + 14).ClearContents
```

La primera línea limpia los datos desde la [Celda A15](#) hasta la [Celda B23](#). El índice 23 de la Celda B se obtiene de sumar el valor de la variable B (9) + 14. Esto ya fue explicado anteriormente. La segunda línea limpia los datos desde la [Celda D15](#) hasta la [Celda F23](#). Si se fijan, las Celdas de la columna C no se limpian porque esta columna contiene fórmulas. Después de esto, solamente nos queda ajustar el número de asiento, que pase al siguiente asiento. Para esto le indicaremos a Excel que establezca el valor de la celda que se llama `no.asiento`, ubicada en la hoja `Asientos`, sumándole 1 al valor que tiene ahorita. El código es el siguiente:

```
Worksheets ("Asientos").Range ("no.asiento") = Worksheets ("Asientos").Range ("no.asiento") + 1
```

Y por último, moveremos el cursor hasta la celda [A14](#). Puede ser cualquier otra.

```
Worksheets ("Asientos").Select
```

```
Range ("A14").Select
```

```
ActiveWorkbook.Save (este es para guardar los cambios en el libro).
```

El código completo debiera quedar de la siguiente manera:


```
Sub guardar_asiento()  
'-----  
'Verificando si el asiento está cuadrado  
'-----  
    If Range("diferencia") <> 0 Then  
        MsgBox "El asiento no puede guardarse porque está descuadrado", vbInformation, "Asiento descuadrado"  
    End  
End If  
'-----  
'Comprobación de campos vacíos y posibles errores en el registro  
'-----  
    If Range("A15") = "" Or Range("B15") = "" Or Range("C15") = "" _  
        Or Range("D15") = "" Or Range("E15") = "" Then  
        MsgBox "No se puede realizar un registro cuando el primer renglón" _  
            & Chr(10) & "del mismo se encuentra vacío. Corrija el error e intente nuevamente" _  
            , vbInformation, "Campos vacíos"  
    End  
End If  
'-----  
'Se realiza el registro del asiento  
'-----  
    Dim A As Long 'Contador de la cantidad de registros que se han agregado en la hoja Registros  
    Dim B As Integer 'Contador de la cantidad de líneas del nuevo asiento que se van a agregar a Registros  
    A = WorksheetFunction.CountIf(Worksheets("Registros").Range("A:A"), "<>" & "")  
    B = WorksheetFunction.CountIf(Worksheets("Asientos").Range("A15", "A" & Rows.Count), "<>" & "")  
    Worksheets("Registros").Range("A" & A + 1) = "Asiento #:" & Worksheets("Asientos").Range("no.asiento")  
    Worksheets("Registros").Range("A" & A + 1).Font.Bold = True
```

```

Worksheets("Asientos").Range("A15", "F" & B + 14).Copy _
Destination:=Worksheets("Registros").Range("B" & A + 2)
Dim i As Integer
Dim j As Integer
j = A + 1 + B
For i = A + 2 To j
    Worksheets("Registros").Range("A" & i) = Worksheets("Asientos").Range("no.asiento")
Next i
Worksheets("Asientos").Range("A15", "B" & B + 14).ClearContents
Worksheets("Asientos").Range("D15", "F" & B + 14).ClearContents
Worksheets("Asientos").Range("no.asiento") = Worksheets("Asientos").Range("no.asiento") + 1
Worksheets("Asientos").Select
Range("A14").Select
ActiveWorkbook.Save
End Sub

```

Ahora, que ya tiene el código completo del botón **Guardar Asiento**. Realice la programación del ícono **Lista de Asientos**. Este ícono al darle clic deberá llevarlo hasta la hoja **“Registros”** y situar el cursor en la celda **“A1”**. Escriba el código de este ícono debajo del código que acabamos de escribir para el ícono anterior. El nombre sugerido para la macro de este ícono es: **mostrar_asientos ()**.

En la hoja **“Registros”** hay un ícono en la parte superior izquierda en la **Celda A1**  Asígnele a este ícono la macro que ya se creó llamada **asientos**.

Esté pendiente en los próximos días, seguiré subiendo material adicional:

Fórmulas con Microsoft Excel	Equivalente con VBA
=Suma()	WorksheetFunction.Sum()
=Consultav() o =Buscarv()	WorksheetFunction.VLookup()
=Sumar.Si()	WorksheetFunction.SumIf()
=Sumar.Si.Conjunto()	WorksheetFunction.SumIfs()
=Contar.Si()	WorksheetFunction.CountIf()
=Contar.Si.Conjunto()	WorksheetFunction.CountIfs()
=Si()	If() o bien: If Expression Then Else Expression End If

Para consultas, escribir a escribeme@radioometepe.com